
ABSTRACT  
Microsoft Excel is still one of the most popular tools for reporting and data exchange. SAS® applications often receive 
input through Excel files and must return data or information in Excel format. This paper illustrates the openness 
between SAS and Microsoft Excel. There are several easy ways to exchange data between SAS and Excel in both 
directions. SAS provides several ways to access data from Excel: the IMPORT procedure and the SAS/ACCESS® 
LIBNAME solution are illustrated. 

Many SAS applications have to provide results in Excel format. The paper illustrates the EXPORT procedure, the 
SAS/ACCESS LIBNAME engine, the %DS2CSV macro, and the power of the Output Delivery System (ODS). ODS 
examples are included using several styles, like MARKUP and the ExcelXP tagset.  

Through the SAS® Add-In for Microsoft Office, Excel users can work directly on SAS data from within their Excel 
sheets. 

 

INTRODUCTION  
Despite of a large range of Business Intelligence tools like SAS, Microsoft Excel is still the most used reporting tool. 
Excel lacks a lot of functionality and flexibility, which can be offered through SAS. Combining the power of SAS with 
Excel increases the productivity of the business users.  

This paper gives a global overview of a wide range of possibilities to work with data or with information:  

• using SAS data in Excel or working with Excel data in SAS. 

• building reports, graphs and analytics in SAS and making the results available in Excel or working in Excel, 
executing tasks and procedures in SAS and returning the results back into Excel. 

An overview will be given of different solutions and a comparison between these solutions is discussed, allowing to 
select the most beneficial one. 

 

ACCESSING DATA 

SAS/ACCESS LIBNAME ENGINE 
A SAS libref can be assigned to non-SAS data. It is not restricted to RDBMSs data like Oracle or DB2, but is also 
available for Microsoft Excel files. The libref points to the “.xls” or “.xlsx” file and every sheet in the file is treated as a 
single data set. Through the LIBNAME engine, SAS will work directly on the Excel data, without the need to build a 
(temporary) SAS data set. 

Consider an Excel file with three sheets. To make this Excel file available to SAS, execute a LIBNAME statement. To 
indicate that the libref refers to non-SAS data, override the automatic default engine, by specifying “EXCEL” as 
engine parameter in the LIBNAME statement. 

 

LIBNAME HOLXLS EXCEL "C:\SGF2009\HOLIDAYS.XLS"; 
 

HOLXLS appears in the list of active libraries in the Explorer window. The small globe on the icon indicates that the 
library points to external, non-SAS data. 

Open the library HOLXLS. You receive the list of data sets (sheets) in the library. Notice that every sheet appears 
twice: once with the name as within Microsoft Excel, and once with the name with a trailing “$”. Excel has “named 
ranges”, allowing data to be placed anywhere in a sheet. SAS reads by default from “named ranges”, but can also 
read from spreadsheets directly. The file with trailing “$” contains the spreadsheet, the other one contains the “named 
ranges”. 
 

                                       
Active Libraries                                                                     Files in Library HOLXLS 

 

Open the data set REVENUE by selecting either “Revenue”’ or “Revenue$” in the Explorer window. The content of 
the Excel sheet is presented. 

 

 
Excel sheet ”Revenue” in SAS 

 

Reading from and Writing to Excel 

Through the SAS/ACCESS LIBNAME engine you can read from and write to Excel files. The Excel SAS/ACCESS 
LIBNAME engine cannot simply replace an existing sheet. To replace a sheet in an Excel file, first remove the sheet, 
using the DATASETS procedure for example. To add data to an existing sheet, specify the option SCAN_TEXT = NO 
in the LIBNAME statement. 

Within the next example, a new sheet named “Continent_DT” is created in the Excel file, starting from another sheet, 
named “Country_DT” in the same Excel file. 

 

PROC SORT DATA = HOLXLS.COUNTRY_DT (KEEP = CONTINENT_CODE) 
           OUT = HOLXLS.CONTINENT_DT NODUPKEY; 
   BY CONTINENT_CODE; 
RUN; 

The result is verified in Microsoft Excel. 

 

 
Excel sheet ”Continent_DT” 

 

Combining Excel Data with SAS Data 

Consider the following data: the Microsoft Excel file - on which the libref HOLXLS is assigned – contains a translation 
table named COUNTRY_DT, with country names and country codes. A SAS table named AT_COUNTRY contains 
revenue data per country code. 

 

   
        Excel sheet ”Country_DT” in Holidays.xls                                             SAS data set AT_COUNTRY 

 

Use these two data sources to build a new SAS data set, showing revenue by country.  

 

PROC SQL; 
   CREATE TABLE COUNTRY_REVENUE AS 
   SELECT COUNTRY_E AS COUNTRY_NAME, 
          BASE_PRICE, CHILD_DISCOUNT, SINGLE_SUPPLEMENT 
   FROM   BIKS.AT_COUNTRY   S, 
          HOLXLS.COUNTRY_DT X 
   WHERE  S.COUNTRY = X.COUNTRY_CODE; 
QUIT; 

SAS data set COUNTRY_REVENUE 

 

THE IMPORT AND EXPORT PROCEDURES 
Access to SAS data in a SAS session or SAS batch program is always faster than access to non-SAS data. When an 
application needs to process the same Excel data multiple times, it might be useful to create a copy of the Excel file 
into a real SAS data set. The IMPORT procedure creates a SAS data set from an Excel sheet; the EXPORT 
procedure creates an Excel sheet from a SAS data set. 

Both procedures can be invoked by executing SAS code, using the Import Wizard or Export Wizard in a SAS session 
or by using the Import Data task in SAS/Enterprise Guide®. 

 

Importing Data into SAS 

Consider the Excel file “BIKS_Info.xls”, containing several sheets. 

 

 
Excel Sheet ”Trip_Item_Codes” 

 

Execute the IMPORT procedure to create a SAS data set WORK.ITEM_DIMENSION from the sheet named 
“Trip_Item_Codes”. 

 

PROC IMPORT OUT  = WORK.ITEM_DIMENSION  
            DBMS = EXCEL REPLACE 
            FILE = "C:\SGF2009\BIKS_INFO.XLS"; 
   SHEET = "TRIP_ITEM_CODES"; 
RUN; 

 

• FILE = names the source Excel file. 

• OUT = names the SAS data set to create. 

• DBMS = EXCEL indicates that the IMPORT procedure treats Excel data. 

• The REPLACE option indicates to override the SAS data set, if it already exists. 

• By default SAS will create the SAS data set, based on the first sheet in the Excel file. Specify a SHEET 
statement to process another sheet.  

 

 
SAS data set ITEM_DIMENSION 

 

Many spreadsheets contain columns with mixed data entry (character values and numeric values). When reading 
data from Excel, SAS scans every column to determine the data type.  

Consider another Excel sheet, containing a column with a mix of numeric and character data (column named 
“BIKS_Code”). 

 

 
Excel sheet ”Country” 

 

Create a SAS data set WORK.COUNTRY_INFO from this Excel sheet, using the IMPORT procedure. 

 

PROC IMPORT OUT  = COUNTRY_INFO  
            DBMS = EXCEL REPLACE 
            FILE = "C:\SGF2009\COUNTRY_INFO.XLS"; 
   SHEET = "COUNTRY"; 
RUN; 

 

Examining the result shows that not all data are imported! For the numeric values in the column “BIKS_Code”, a 
missing value appears in the SAS data set.  

 

 
SAS data set COUNTRY_INFO 

To solve this problem / error, you can specify a MIXED = YES statement in the IMPORT procedure. Notice that the 
option MIXED = YES can also be specified in a LIBNAME statement, using the Excel engine. 

 

PROC IMPORT OUT  = COUNTRY_INFO  
            DBMS = EXCEL REPLACE 
            FILE = "C:\SGF2009\COUNTRY_INFO.XLS"; 
   SHEET = "COUNTRY"; 
   MIXED = YES; 
RUN; 

 

 
SAS data set COUNTRY_INFO 

 

Exporting SAS Data Sets to Excel 

The EXPORT procedure enables creating Excel data from within a SAS session or SAS program.  

Consider the SAS dataset BIKS.REVENUE_FACTS. Use the EXPORT procedure to build a sheet named “Global 
Revenue” in the Excel file named “Revenue_Facts.xls” from this SAS data set.  

 

 
SAS Table BIKS.REVENUE_FACTS (partial) 

 

PROC EXPORT DATA = BIKS.REVENUE_FACTS 
            DBMS = EXCEL REPLACE 
            FILE = "C:\SGF2009\REVENUE_FACTS.XLS"; 
   SHEET = "GLOBAL REVENUE"; 
RUN; 

 

• FILE = names the Excel file to create or to modify. 

• REPLACE indicates to override the sheet, when it already exists. 

• By default the sheet receives the name of the SAS data set as label. Use the SHEET = statement to assign 
another label to the sheet. 

• SAS date values are presented as dates in Excel. 

Excel file ”Revenue_Facts.xls” 

 

THE %DS2CSV MACRO 
Microsoft Excel is not available on platforms like Unix, but still users want to create output that can be treated in Excel 
on their PC. SAS provides the macro %DS2CSV to build “.csv” (comma separated values) output for Excel. The 
macro has a set of named parameter, with default values. You need to specify at least the name of the source data 
set (DATA = parameter), the output file name (CSVFILE = parameter) and the option RUNMODE = B. 

Use the %DS2CSV macro to create a “.csv” file from the SAS data set BIKS.HOLIDAYS. 

 

%DS2CSV (CSVFILE  = "C:\SGF2009\DS_HOLIDAYS.CSV", 
         DATA     = BIKS.HOLIDAYS, 
         RUNMODE  = B); 

 

Partial file ”DS_Holidays.csv” 

 
"Trip_Description","Trip_Type","Continent","Country","Trip_Revenue","Tour_Traveller_
Count","Tour_Score" 
"Discovery of China","Tour","Asia","China","10380.5","7","57" 
"Big Tour of China","Tour","Asia","China","6642","4","79" 
"Discovery of China","Tour","Asia","China","11586","8","75" 
"Big Tour of China","Tour","Asia","China","7072","4","82" 
"Discovery of China","Tour","Asia","China","5370","4","52" 
"Big Tour of Cuba","Tour","South America","Cuba","6542","4","58" 
"Discovery of Cuba","Tour","South America","Cuba","10881","6","39" 
 

 

 

The following example illustrates a few other parameters: use the SAS data set named BIKS.HOLIDAYS to create the 
Excel file named “DS_Holidays.csv”. Modify the default output through parameters in the macro call: 

• When the sheet already exists, override its content with new data. (OPENMODE = option)  

• By default the first line in the output file contains the name of the variables. Within this example this first line is 
not generated. (COLHEAD = option)  

• By default all variables of the SAS data set are written to the output file, in the same sequence as their sequence 
in the observations. Use the VAR = parameter to select the variables to export and to define their sequence in 
the output file. 

• By default all observations are exported. Specify a WHERE = parameter to subset the data to records with the 
value “Thailand” for the variable COUNTRY. 

%DS2CSV (CSVFILE  = "C:\SGF2009\DS_HOLIDAYS.CSV", 
         OPENMODE = REPLACE, 
         COLHEAD  = N, 
         VAR      = TRIP_DESCRIPTION TRIP_TYPE TRIP_REVENUE, 
         DATA     = BIKS.HOLIDAYS, 
         WHERE    = COUNTRY = 'Thailand', 
         RUNMODE  = B); 

 

Partial file ”DS_Holidays.csv” 

"Discovery of Thailand","Tour","7750" 
"Discovery of Thailand","Tour","5939" 
"Big Tour of Thailand","Tour","8231.5" 
"Discovery of Thailand","Tour","5151" 
"Big Tour of Thailand","Tour","8859" 
"Discovery of Thailand","Tour","5342.5" 
"Discovery of Thailand","Tour","7197" 
"Discovery of Thailand","Tour","7164" 
"Discovery of Thailand","Tour","10096" 
"Discovery of Thailand","Tour","15161" 
"Discovery of Thailand","Tour","6302" 

 

 

CREATING RESULTS IN EXCEL FORMAT 
Instead of exchanging data between SAS and Excel, reports, graphs or analysis results can be prepared in SAS and 
published in Excel. The business users can concentrate on analyzing the results in Excel instead of spending (too 
much) time on working on the lay-out of the information. Moreover, SAS is more self-documenting than Excel, which 
makes maintenance a lot easier. 

 

THE OUTPUT DELIVERY SYSTEM (ODS) 
By default SAS generates output to the LISTING destination. This output uses mono-space fonts, and was designed 
for traditional line-printers. The Output Delivery System (ODS) introduces more flexibility in output generation with a 
wide range of formatting options. ODS enables generating output that is treated by other applications like Excel, 
Acrobat, Web Browsers, etc.  

ODS allows creating output to different destinations. To start building output for a specific destination, execute: 

   ODS destination <FILE = output_file> <addional_parameters>; 
 

To stop generating output to a specific destination, execute: 

   ODS destination CLOSE; 
 

• destination names the output destination. Valid values include: RTF, PDF, HTML, PRINTER, LISTING, CSV, 
MARKUP, … 

• FILE = output_file names the physical name of the output file. This can be replaced by a FILEREF = option. 

• additional_parameters (optional) specify for example a style to use. A style is a template describing the formatting 
of the output. The STYLE = option can refer to SAS supplied styles or user-defined styles. 

 

The following examples generate two reports, for which the code is defined in a macro:  

%MACRO TABRPT; 
   TITLE "Trip Revenue"; 
   PROC TABULATE DATA = BIKS.HOLIDAYS; 
      CLASS CONTINENT TRIP_TYPE; 
      VAR   TRIP_REVENUE; 

      TABLE CONTINENT = " ", 
            TRIP_TYPE = " " * TRIP_REVENUE = " " * SUM = " " * F = COMMA11.; 
   RUN; 
%MEND; 
 
%MACRO REPRPT; 
   TITLE "Revenue Analysis per Country"; 
   PROC REPORT DATA = BIKS.HOLIDAYS_AG NOWD; 
      WHERE TRIP_TYPE LIKE "T%"; 
      COLUMNS COUNTRY TRIP_REVENUE NBR_OF_TRIPS AVG_REVENUE; 
      DEFINE COUNTRY / GROUP; 
      DEFINE TRIP_REVENUE / SUM FORMAT = COMMA8.; 
      DEFINE NBR_OF_TRIPS / SUM; 
      DEFINE AVG_REVENUE  / COMPUTED FORMAT = COMMA8.; 
      COMPUTE AVG_REVENUE; 
         AVG_REVENUE = TRIP_REVENUE.SUM / NBR_OF_TRIPS.SUM; 
      ENDCOMP; 
   RUN; 
%MEND; 

 

The ODS CSV Destination 

Use the ODS CSV destination to create two reports, and examine the result in Excel. 

 

ODS CSV FILE = "C:\SGF2009\BIKS_CSV.CSV";  
%TABRPT; 
%REPRPT;   
ODS CSV CLOSE;  

 

• The ODS CSV statement opens the CSV destination and names the output file. All statements following the ODS 
statement will generate their output to that destination, until the ODS CSV CLOSE statement is encountered.  

• The ODS CSV CLOSE statement ends the process of output generation to this destination. 

 

 
ODS CSV output in Excel 

 

• Notice that the result of both procedures is written to one Excel sheet. 

• Notice that the result of the TITLE statements is not written to the CSV file. Using ODS CSVALL instead of ODS 
CSV will add the titles.  

The ODS MARKUP Destination 

The ODS MARKUP destination is used to create XML from SAS output. This XML can be used in a Web Browser or 
in Microsoft Excel. The TAGSET = option in the ODS MARKUP statement specifies the value of a tagset. A tagset is 
a template that defines how to create a markup language output type. Valid values include HTML, CSV, CSVALL.  

The example below illustrates the creation of a “.csv” file, containing two reports – with titles – within one sheet. 

 

ODS MARKUP TAGSET = CSVALL FILE = "C:\SGF2009\BIKS_CSVALL.CSV";  
%TABRPT; 
%REPRPT;   
ODS MARKUP CLOSE;  

 

 
ODS MARKUP output in Excel 

 

The ODS Tagsets.ExcelXP Destination 

The ODS TAGSETS.EXCELXP destination allows generating multiple worksheets per Excel file. This ODS 
destination generates output in the SpreadsheetML format. The EXCELXP destination allows additional parameters 
to improve the layout of the output.  

The examples for the ODS TAGSETS.EXCELXP build a report using the REPORT procedure, executing the following 
statements. A BY statement indicates to build a report for each value of the variable “TRIP_TYPE”: 

 

%MACRO RPRT; 
TITLE "Revenue Analysis per Country"; 
   PROC REPORT DATA = BIKS.HOLIDAYS_AG NOWD; 
      BY TRIP_TYPE; 
      COLUMNS CONTINENT TRIP_REVENUE NBR_OF_TRIPS AVG_REVENUE; 
      DEFINE CONTINENT / GROUP; 
      DEFINE TRIP_REVENUE / SUM FORMAT = COMMA8.; 
      DEFINE NBR_OF_TRIPS / SUM; 
      DEFINE AVG_REVENUE  / COMPUTED FORMAT = COMMA8.; 
      COMPUTE AVG_REVENUE; 
          AVG_REVENUE = TRIP_REVENUE.SUM / NBR_OF_TRIPS.SUM; 
      ENDCOMP; 
   RUN; 
%MEND; 

 

Create an Excel sheet using the ODS TAGSETS.EXCELXP destination. As a result of the BY statement (with two 
distinct values for TRIP_TYPE) two sheets are created, one for each value of the BY-variable. The BY-line is shown 

in each sheet. Through the OPTIONS option, the name of the first sheet is specified (“Revenue”). Subsequent sheets 
receive the same name, followed by a sequence number.  

 

ODS TAGSETS.EXCELXP FILE  = "C:\SGF2009\EXCELXP.XLS"  
                    STYLE = SASWEB 
        OPTIONS (SHEET_NAME = 'Revenue'); 
%RPRT; 
ODS TAGSETS.EXCELXP CLOSE; 

 

 
ODS Tagsets.ExcelXP output in Excel 

 

Modify the options in the ODS statement. The SHEET_INTERVAL = “NONE” option indicates that, although a BY 
statement is used, all output must be created in one sheet.  

 

ODS TAGSETS.EXCELXP FILE  = "C:\SGF2009\EXCELXP.XLS"  
                    STYLE = SASWEB  
        OPTIONS (SHEET_INTERVAL = "NONE"  
                 SHEET_NAME     = "Revenue"); 
%RPRT; 
ODS TAGSETS.EXCELXP CLOSE; 
 

 

 
ODS Tagsets.ExcelXP output in Excel 

 

A final example on the ODS TAGSETS.EXCELXP destination specifies options to:  

• Allow users to filter on each column within Excel (the AUTOFILTER = option). 

• Keep the header lines presented when scrolling down in the Excel sheet. 

ODS TAGSETS.EXCELXP FILE  = "C:\SGF2009\EXCELXP.XLS"  
                    STYLE = SASWEB  
        OPTIONS (AUTOFILTER     = "ALL"  
                 FROZEN_HEADERS = "YES"  
                 SHEET_NAME     = "Revenue"); 
TITLE "Revenue Analysis per Country"; 
PROC PRINT DATA = BIKS.HOLIDAYS_AG NOOBS; 
   VAR TRIP_TYPE CONTINENT COUNTRY TRIP_REVENUE; 
   FORMAT TRIP_REVENUE COMMA8.; 
RUN;   
ODS TAGSETS.EXCELXP CLOSE; 

 

Review the output in Excel and scroll down three lines. Notice that the headers are frozen. 

 

 
ODS Tagsets.ExcelXP output in Excel 

 

Use the drop down icons to filter the Excel on TRIP_TYPE = “Tour” and CONTINENT = “Asia”. 

 

 
ODS Tagsets.ExcelXP output in Excel  

 

 

The ODS HTML Destination 

The ODS HTML destination is by default used to build output to view in a Web Browser. By specifying the “.xls” or 
“.xlsx” file type suffix in the file name, an Excel output is created.  

The following example illustrates the creation of a PROC REPORT output, enhanced with exception highlighting by 
using formats. 

PROC FORMAT; 
   VALUE BACK    0 - 550000 = "RED" 
              650000 - HIGH = "GREEN" 
                      OTHER = "WHITE"; 
 
ODS HTML FILE  = "C:\SGF2009\REPORT.XLS"  
         STYLE = SASDOCPRINTER; 
PROC REPORT DATA = BIKS.HOLIDAYS NOWD; 
   COLUMN CONTINENT COUNTRY TRIP_REVENUE; 
   DEFINE CONTINENT / GROUP; 
   DEFINE COUNTRY   / GROUP; 
   DEFINE TRIP_REVENUE / ANALYSIS SUM FORMAT = COMMA12. "Revenue"  
                         STYLE (COLUMN) = {BACKGROUND = BACK.}; 
RUN; 
ODS HTML CLOSE; 

 

 
ODS HTML to Excel file, with Exception Highlighting 

 

The STYLE (COLUMN) option specifies the background color of the TRIP_REVENUE column. The calculated value 
is translated using a format, named BACK (translating 2016136 into “GREEN” for example). As a result the Excel file 
is created with a dynamic color layout, based on the cell values. 

 

 

SAS® ADD-IN FOR MICROSOFT OFFICE  
SAS Add-In for Microsoft Office is an add-in that enables users to transparently leverage the power of SAS directly 
from Excel, Word or Powerpoint. It allows Excel users to access directly data in SAS data sets or provides Excel 
users access to SAS procedures to be executed on SAS data sets, returning results into their Excel files. 

 

ACCESSING SAS DATA IN EXCEL 
SAS Add-In for Microsoft Office is capable of accessing very large data sources, eventually bypassing the data size 
limit of Excel.  

SAS Add-In for Microsoft Office is installed as an integrated menu option within Microsoft Office applications (Excel, 
Word, Powerpoint). The traditional standard menu is extended with an option named “SAS”. 

Excel with SAS® Add-In for Microsoft Office installed 

 

SAS tasks are also available through two SAS toolbars. 

 

                  
       SAS Data Analysis Toolbar                                                                    SAS Analysis Tools Toolbar 

 

 

Example: Working with SAS Data Sets 

The SAS data set TRIP_FACTS contains 12 variables and more than 150.000 observations. Analyze this SAS data 
set in Excel. 

• Select the Open icon on the SAS toolbar.                      

• Navigate to the location of the SAS data and select the data set. The list of variables of the data set is shown in 
the right list box. By default all variables are selected.  

• Deselect the variables that are not required to be shown in Excel. 

• Select the Open button to display the data in Excel. 

SAS data set TRIP_FACTS in Excel (partial) 

 

When the data set exceeds the data size limit of the Excel version, you can navigate easily through portions of the 
data set, using the SAS toolbar.  

 

 
 

Navigating to specific parts of the SAS data set, or to a specific observation is also available from the pull-down 
menu:  

• Select “SAS” on the pull-down menu. 

• Select “Navigate SAS Data Source” from the drop-down list. 

• Select an item from the pop-up list. 

 

         
Drop-down lists from ”SAS” pull-down menu item 

Filtering SAS Data Sets 

A point-and-click interface allows building filters, applied on the SAS data set before bringing the data into Excel. 
Afterwards, Excel filters can be added to further subset the data in Excel.  

 

• Select the Filter icon on the SAS Data Analysis toolbar.             

• Use the drop-down arrows and selection lists to build the filter expression. 

 

 
 

Advantages of filtering the data in SAS include: 

• Apply filters on variables that are not transferred to Excel. 

• Build complex filters using AND and OR relations. 

• The Advanced Edit… button opens the Advanced Expression Editor window enabling to type any filter 
expression. 

• The filters are applied on SAS data sets in SAS, without any size limits. 

 

 

Sorting SAS Data Sets 

Data can be sorted before presenting that data in Excel. The sort is executed in SAS and can be extended by a sort 
in Excel. 

• Select the Sort icon on the SAS Data Analysis toolbar.         

• Specify one or more variables to sort on and define the sort as ascending or descending.  

• Select the Preview Data check box to get a preview of the result. The size of the subset to display in the preview 
can easily be adapted. 

EXECUTING SAS TASKS IN EXCEL 
Business users can exploit the full power of SAS to build reports, charts, data analysis results on their data from 
within their familiar Excel environment. Almost all procedures are available through a point-and-click environment. 

 

Example 

The SAS data set HOLIDAYS contains information about holiday trips from a travel agency. Use this SAS data set to 
build a tabular report, showing revenue figures, calculated per continent and per trip type. Execute this task from 
within an Excel sheet, and return the result in the same Excel file, in a new sheet. 

 

• Open the SAS data set HOLIDAYS in Excel. 

 

 
SAS data set HOLIDAYS in Excel 

• Select My SAS Favorites from the SAS Analysis Tools toolbar  

•  

 
 

• A list of standard SAS tasks and user-defined SAS tasks is shown. (Notice that this list is comparable with the list 
that you receive in SAS Enterprise Guide). Select Summary Tables. 

 

 
 

• Select Task Roles in the left list box and define variables for analysis and for classification. Optionally specify 
additional options for each variable. 

 

 
 

• Select Summary Tables in the left list box. Use drag-and-drop to draw the layout of the final tabular report. 
Optionally request totals, subtotals, percentages, etc. 

 

Foundations and FundamentalsSAS Global Forum 2009

 



19 

 
 

• Select the Run button to build the SAS code and execute this code. The result is shown in a new sheet in the 
Excel file. 

 
PROC TABULATE Output in Excel 

 

 

CONCLUSION 
Users have several possibilities to work with Excel data in SAS. The following table compares three methods: 

 

Method Advantages Disadvantages 

SAS/Access LIBNAME Engine Easy 

Transparent 

Performance: when data are required in 
several tasks, data are each time read from 
Excel 

IMPORT / EXPORT Procedure A SAS copy is created, with 
positive impact on performance. 

A SAS copy is created, resulting in extra 
space requirements 

%DS2CSV macro Available on platforms like Unix, 
that do not run Microsoft Office 

More complex coding 

 

SAS Add-In for Microsoft Office allows working with (large) SAS data sets directly in Excel. Benefits include:  

• Subset, sort the data in SAS. 

• Navigate through large tables. 

 

SAS ODS offers several possibilities to create information by means of reports and graphs to be displayed in Excel. 
Additional features and parameters enable passing display options to Excel, like freezing columns or exception 
highlighting. 

 

Finally the SAS Add-In for Microsoft Office offers the full power of SAS from within Excel, to build documented, 
powerful applications.  

